PL/SQL Script Line with Only a Dot...What is the purpose/function of the dot? I have inherited a pl/sql script from a previous dba that worked at my company. It is used to "clean up" failed distributed transactions. The script is easy enough to understand, except for one line that only contains a dot, "." I have not seen this before...what is the purpose/function of the dot line?

set define on
set scan on
set verify off
set serveroutput on

undefine IncidentNumber

spool &amp;&amp;IncidentNumber

connect / as sysdba
whenever sqlerror continue
set serveroutput on size 1000000
set linesize 200
DECLARE
   /* &amp;&amp;IncidentNumber */
   pragma autonomous_transaction;
   l_sql_statement       VARCHAR2(255);
   l_fdist_found         BOOLEAN := FALSE;
   --
   CURSOR cfdist IS
     SELECT local_tran_id tranid, upper(state) state
     FROM dba_2pc_pending;
   --
   BEGIN
      --
      FOR rfdist IN cfdist LOOP
          l_fdist_found := TRUE;
          if rfdist.state = 'PREPARED' then
              l_sql_statement := 'ROLLBACK FORCE '''||rfdist.tranid||''''||'/*&amp;&amp;IncidentNumber*/';
              DBMS_OUTPUT.PUT_LINE(l_sql_statement);
              EXECUTE IMMEDIATE l_sql_statement;
              COMMIT;
          end if;
          DBMS_OUTPUT.PUT_LINE('Executing DBMS_TRANSACTION.PURGE_LOST_DB_ENTRY('||rfdist.tranid||')');
          DBMS_TRANSACTION.PURGE_LOST_DB_ENTRY(rfdist.tranid)/*&amp;&amp;IncidentNumber*/;
          COMMIT;
          IF l_fdist_found = TRUE THEN
              DBMS_OUTPUT.PUT_LINE('Rolled back distributed transaction id '|| rfdist.tranid);
          END IF;
      END LOOP;
      --
      IF l_fdist_found = FALSE THEN
          DBMS_OUTPUT.PUT_LINE('No failed distributed transactions were found.');
      END IF;
      --
     EXCEPTION
       WHEN OTHERS THEN
           RAISE_APPLICATION_ERROR(-20001, 'Error:'||SQLCODE||':'||SQLERRM);
   END;
.
/
A line with only a dot, ".", appears to be a SQL*Plus command to end the input and save it in the buffer without executing it. 

SQL&gt;
SQL&gt;
SQL&gt; DECLARE
  2     qty1 number;
  3  BEGIN
  4     select count(*) into qty1 from dba_roles;
  5     dbms_output.put_line('Number of roles: '||to_char(qty1));
  6  END;
  7  .
SQL&gt;
SQL&gt;
SQL&gt; LIST
  1  DECLARE
  2     qty1 number;
  3  BEGIN
  4     select count(*) into qty1 from dba_roles;
  5     dbms_output.put_line('Number of roles: '||to_char(qty1));
  6* END;
SQL&gt;
SQL&gt;
SQL&gt;
SQL&gt;
SQL&gt;
SQL&gt; DECLARE
  2     qty2 number;
  3  BEGIN
  4     select count(*) into qty2 from dba_procedures;
  5     dbms_output.put_line('Number of procedures: '||to_char(qty2));
  6  END;
  7  .
SQL&gt;
SQL&gt;
SQL&gt;
SQL&gt; LIST
  1  DECLARE
  2     qty2 number;
  3  BEGIN
  4     select count(*) into qty2 from dba_procedures;
  5     dbms_output.put_line('Number of procedures: '||to_char(qty2));
  6* END;
SQL&gt;
SQL&gt;
SQL&gt;
SQL&gt; /
Number of procedures: 26744

PL/SQL procedure successfully completed.

SQL&gt;
SQL&gt;